
    TinyElixirStripe

Warning
This library is still experimental! It's thoroughly tested in ExUnit, but that's it.
Stripe doesn't provide an official Elixir SDK, and maintaining a full-featured SDK has proven to be a challenge. As I see it, this is because the Elixir community is pretty senior-driven. People have no problem rolling their own integration with the incredible Req library. 
This library is an attempt to wrap those community learnings in an easy-to-use package modeled after patterns set forth in Dashbit's SDKs with Req: Stripe article by Wojtek Mach. 
My hope is that this should suffice for 95% of all apps that need to integrate with Stripe, and that the remaining 5% of use cases have a built-in escape hatch with Req. 
Features
	Simple API Client built on Req with automatic ID prefix recognition
	Webhook Handler DSL using Spark for clean, declarative webhook handling
	Automatic Signature Verification for webhook security
	Code Generators powered by Igniter for zero-config setup
	Sync with Stripe to keep your local handlers in sync with your Stripe dashboard

Installation
Using Igniter (Recommended)
The fastest way to install is using the Igniter installer:
mix igniter.install tiny_elixir_stripe

This will:
	Add the dependency to your mix.exs
	Replace Plug.Parsers with TinyElixirStripe.ParsersWithRawBody in your Phoenix endpoint (required for webhook signature verification)
	Create a StripeWebhookHandlers module for defining event handlers
	Generate a StripeWebhookController in your Phoenix app
	Add the webhook route to your router (default: /webhooks/stripe)
	Configure .formatter.exs for DSL formatting support

Then configure your Stripe credentials in config/runtime.exs:
config :tiny_elixir_stripe,
  stripe_api_key: System.get_env("YOUR_STRIPE_KEY_ENV_VAR"),
  stripe_webhook_secret: System.get_env("YOUR_WEBHOOK_SECRET_ENV_VAR")
Manual Installation
If you prefer not to use Igniter, add to your mix.exs:
def deps do
  [
    {:tiny_elixir_stripe, "~> 0.1"}
  ]
end
Then follow the Manual Setup instructions below.
Changing the Webhook Path
The default webhook path is /webhooks/stripe. If you need to change it later:
mix tiny_elixir_stripe.set_webhook_path /new/webhook/path

Manual Installation (without Igniter)
If you prefer not to use Igniter, you'll need to manually:
	Replace Plug.Parsers in your endpoint (lib/my_app_web/endpoint.ex):

# Replace this:
plug Plug.Parsers,
  parsers: [:urlencoded, :multipart, :json],
  pass: ["*/*"],
  json_decoder: Jason

# With this:
plug TinyElixirStripe.ParsersWithRawBody,
  parsers: [:urlencoded, :multipart, :json],
  pass: ["*/*"],
  json_decoder: Jason
	Create a webhook handler module (lib/my_app/stripe_webhook_handlers.ex):

defmodule MyApp.StripeWebhookHandlers do
  use TinyElixirStripe.WebhookHandler

  # Define your handlers here (see examples below)
end
	Create a webhook controller (lib/my_app_web/stripe_webhook_controller.ex):

defmodule MyAppWeb.StripeWebhookController do
  use TinyElixirStripe.WebhookController,
    handler: MyApp.StripeWebhookHandlers
end
	Add the route to your router (lib/my_app_web/router.ex):

scope "/webhooks" do
  post "/stripe", MyAppWeb.StripeWebhookController, :create
end
	Add formatter config to .formatter.exs:

[
  import_deps: [:tiny_elixir_stripe],
  # ... rest of config
]
Handling Stripe Webhooks
TinyElixirStripe provides a clean DSL for handling webhook events. When Stripe sends a webhook to your endpoint, the controller automatically:
	Verifies the webhook signature using your signing secret
	Parses the event
	Dispatches it to the appropriate handler

Function Handlers
Define inline handlers for simple event processing:
defmodule MyApp.StripeWebhookHandlers do
  use TinyElixirStripe.WebhookHandler

  handle "customer.created", fn event ->
    customer_id = event["data"]["object"]["id"]
    email = event["data"]["object"]["email"]
    
    # Your business logic here
    MyApp.Customers.create_from_stripe(customer_id, email)
    
    :ok
  end

  handle "customer.updated", fn event ->
    # Handle customer updates
    :ok
  end

  handle "invoice.payment_succeeded", fn event ->
    # Handle successful payments
    :ok
  end
end
Module Handlers
For more complex event processing, use separate modules:
defmodule MyApp.StripeWebhookHandlers do
  use TinyElixirStripe.WebhookHandler

  handle "customer.subscription.created", MyApp.StripeWebhookHandlers.SubscriptionCreated
  handle "customer.subscription.updated", MyApp.StripeWebhookHandlers.SubscriptionUpdated
  handle "customer.subscription.deleted", MyApp.StripeWebhookHandlers.SubscriptionDeleted
end
defmodule MyApp.StripeWebhookHandlers.SubscriptionCreated do
  @moduledoc """
  Handles subscription creation events.
  """

  def handle_event(event) do
    subscription = event["data"]["object"]
    customer_id = subscription["customer"]
    
    # Complex business logic
    with {:ok, user} <- MyApp.Users.find_by_stripe_customer(customer_id),
         {:ok, _subscription} <- MyApp.Subscriptions.create(user, subscription) do
      :ok
    else
      {:error, reason} ->
        {:error, reason}
    end
  end
end
Generating Handlers
Use the generator to quickly scaffold handlers:
# Generates a function handler
mix tiny_elixir_stripe.gen.handler customer.created

# Generates a module handler
mix tiny_elixir_stripe.gen.handler customer.subscription.created --handler-type module

# Generates with custom module name
mix tiny_elixir_stripe.gen.handler charge.succeeded --handler-type module --module MyApp.Payments.ChargeHandler

The generator will:
	Validate the event name against supported Stripe events
	Create the handler in your WebhookHandler module
	Generate a separate module file for module handlers

Syncing with Stripe
Keep your local handlers in sync with your Stripe webhook configuration:
mix tiny_elixir_stripe.sync_webhook_handlers

This task will:
	Fetch all webhook endpoints from your Stripe account
	Extract all enabled events
	Compare them with your existing handlers
	Generate stub handlers for any missing events

Options:
	--handler-type function|module|ask - Choose handler type for all missing events
	--skip-confirmation or -y - Skip prompts and generate all handlers
	--api-key or -k - Specify Stripe API key (otherwise uses config or prompts)

Example output:
Fetching webhook endpoints from Stripe...

Found 1 webhook endpoint(s):
  • https://myapp.com/webhooks/stripe (5 events)

Collecting all enabled events...

Events configured in Stripe:
  ✓ customer.created (handler exists)
  ✗ customer.updated (missing)
  ✗ invoice.payment_succeeded (missing)
  ✓ subscription.created (handler exists)
  ✗ subscription.deleted (missing)

Found 3 missing handler(s) out of 5 total Stripe event(s).

Generate handlers for missing events? (y/n) y

What type of handlers would you like to generate?
1. function
2. module
3. ask
> 1

Generating handlers...
  • customer.updated (function handler)
  • invoice.payment_succeeded (function handler)
  • subscription.deleted (function handler)

✓ Done! Generated 3 new handler(s).
Calling the Stripe API
The TinyElixirStripe.Client module provides a simple CRUD interface for interacting with the Stripe API, built on Req.
Basic Usage
alias TinyElixirStripe.Client

# Fetch a customer by ID
{:ok, response} = Client.read("cus_123")
customer = response.body

# List customers with pagination
{:ok, response} = Client.read(:customers, limit: 10, starting_after: "cus_123")
customers = response.body["data"]

# Create a customer
{:ok, response} = Client.create(:customers, %{
  email: "customer@example.com",
  name: "Jane Doe",
  metadata: %{user_id: "12345"}
})

# Update a customer
{:ok, response} = Client.update("cus_123", %{
  name: "Jane Smith",
  metadata: %{premium: true}
})

# Delete a customer
{:ok, response} = Client.delete("cus_123")
Automatic ID Recognition
The client automatically recognizes Stripe ID prefixes:
Client.read("cus_123")      # => /customers/cus_123
Client.read("sub_456")      # => /subscriptions/sub_456
Client.read("price_789")    # => /prices/price_789
Client.read("product_abc")  # => /products/product_abc
Client.read("inv_xyz")      # => /invoices/inv_xyz
Client.read("evt_123")      # => /events/evt_123
Client.read("cs_test_abc")  # => /checkout/sessions/cs_test_abc
Supported Entity Types
Use atoms for entity types when creating or listing:
Client.create(:customers, %{email: "test@example.com"})
Client.create(:subscriptions, %{customer: "cus_123", items: [%{price: "price_abc"}]})
Client.create(:products, %{name: "Premium Plan"})
Client.create(:prices, %{product: "prod_123", unit_amount: 1000, currency: "usd"})
Client.create(:checkout_sessions, %{mode: "payment", line_items: [...]})

Client.read(:customers, limit: 100)
Client.read(:subscriptions, customer: "cus_123")
Client.read(:invoices, status: "paid")
Bang Functions
Use ! versions to raise on errors:
# Raises RuntimeError on failure
response = Client.read!("cus_123")
customer = Client.create!(:customers, %{email: "test@example.com"})
Advanced Usage with Req
Since the client is built on Req, you can access the full Req API:
# Direct Req request with custom options
{:ok, response} = Client.request("/charges/ch_123", retry: :transient)

# Or build a custom client
client = Client.new(receive_timeout: 30_000)
{:ok, response} = Req.get(client, url: "/customers/cus_123")
Testing
Configure Req.Test for testing:
# config/test.exs
config :tiny_elixir_stripe,
  req_options: [plug: {Req.Test, TinyElixirStripe}]
In your tests:
test "creates a customer" do
  Req.Test.stub(TinyElixirStripe, fn conn ->
    Req.Test.json(conn, %{
      id: "cus_test_123",
      email: "test@example.com",
      object: "customer"
    })
  end)

  {:ok, response} = Client.create(:customers, %{email: "test@example.com"})
  assert response.body["id"] == "cus_test_123"
end
Configuration
# config/config.exs
config :tiny_elixir_stripe,
  stripe_api_key: System.get_env("STRIPE_SECRET_KEY"),
  stripe_webhook_signing_secret: System.get_env("STRIPE_WEBHOOK_SECRET")

# config/test.exs  
config :tiny_elixir_stripe,
  req_options: [plug: {Req.Test, TinyElixirStripe}]

    Changelog

All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
[0.0.2] - 2025-12-18
Added
	Docs with HexDocs
	This changelog
	Hex publish configurations

0.0.1 - 2025-12-18
Added
	Initial release
	Stripe API client built on Req with automatic ID prefix recognition
	Webhook handler DSL using Spark
	Automatic webhook signature verification
	mix tiny_elixir_stripe.install - Igniter-powered installation task
	mix tiny_elixir_stripe.gen.handler - Generate webhook event handlers
	mix tiny_elixir_stripe.sync_webhook_handlers - Sync handlers with Stripe dashboard
	mix tiny_elixir_stripe.set_webhook_path - Change webhook endpoint path
	mix tiny_elixir_stripe.update_supported_events - Update supported Stripe events list
	Support for both function and module-based webhook handlers
	CRUD operations for common Stripe entities (customers, subscriptions, products, prices, etc.)
	Comprehensive test coverage
	Full documentation and guides



  

    
TinyElixirStripe.WebhookHandler.Dsl.Handler 
    



      
Represents a webhook event handler.

      




  

    
TinyElixirStripe.WebhookHandler.Transformers.GenerateHandleEvent 
    



      
Transformer that generates handle_event/2 functions for webhook handlers.

TinyElixirStripe 
    



      
A minimal Stripe SDK for Elixir with webhook handling, built on Req and Spark.
Warning
This library is still experimental! It's thoroughly tested in ExUnit, but that's it.
TinyElixirStripe provides:
	Simple API Client - Built on Req with automatic ID prefix recognition
	Webhook Handler DSL - Using Spark for clean, declarative webhook handling
	Automatic Signature Verification - For webhook security
	Code Generators - Powered by Igniter for zero-config setup
	Sync with Stripe - Keep your local handlers in sync with your Stripe dashboard

Quick Start
Install with Igniter:
mix tiny_elixir_stripe.install
Configure your API key:
config :tiny_elixir_stripe,
  stripe_api_key: System.get_env("STRIPE_SECRET_KEY")
Handle webhook events:
defmodule MyApp.StripeWebhookHandlers do
  use TinyElixirStripe.WebhookHandler

  handle "customer.created", fn event ->
    # Your logic here
    :ok
  end
end
Call the Stripe API:
alias TinyElixirStripe.Client

{:ok, response} = Client.read("cus_123")
{:ok, response} = Client.create(:customers, %{email: "test@example.com"})
For full documentation, see the README.

      




  

    
TinyElixirStripe.Client 
    



      
A minimal Stripe API client built on Req.
This module provides a simple interface for interacting with the Stripe API,
covering the most common operations while maintaining an escape hatch to Req
for advanced use cases.
Configuration
Configure your Stripe API key in your application config:
config :tiny_elixir_stripe,
  stripe_api_key: "sk_test_..."
For testing, configure the client to use Req.Test:
config :tiny_elixir_stripe,
  req_options: [plug: {Req.Test, TinyElixirStripe}]
CRUD Operations
The client supports standard CRUD operations with a consistent API:
Read Operations
Fetch individual resources by ID (string):
Client.read("cus_123")
# => {:ok, %Req.Response{body: %{"id" => "cus_123", ...}}}
List resources by entity type (atom):
Client.read(:customers, limit: 10)
# => {:ok, %Req.Response{body: %{"object" => "list", "data" => [...]}}}
Create Operations
Create resources using entity type atoms:
Client.create(:customers, %{email: "user@example.com"})
# => {:ok, %Req.Response{body: %{"id" => "cus_new", ...}}}
Update Operations
Update resources by ID:
Client.update("cus_123", %{name: "New Name"})
# => {:ok, %Req.Response{body: %{"id" => "cus_123", "name" => "New Name"}}}
Delete Operations
Delete resources by ID:
Client.delete("cus_123")
# => {:ok, %Req.Response{body: %{"id" => "cus_123", "deleted" => true}}}
Supported Entity Types
The following entity types are supported (as atoms):
	:customers - Customer objects
	:products - Product objects
	:prices - Price objects
	:subscriptions - Subscription objects
	:invoices - Invoice objects
	:events - Event objects
	:checkout_sessions - Checkout Session objects

Error Handling
All operations return {:ok, response} or {:error, reason} tuples:
Client.read(:invalid_entity)
# => {:error, :unrecognized_entity_type}

# HTTP errors return the response with error details
Client.read("cus_nonexistent")
# => {:error, %Req.Response{status: 404, body: %{"error" => ...}}}
Bang Functions
Each CRUD function has a bang version (read!/2, create!/3, update!/3, delete!/2)
that raises a RuntimeError instead of returning error tuples:
# Raises on error
response = Client.read!("cus_123")
customer = response.body

# Raises RuntimeError: "Unrecognized entity type: :invalid"
Client.read!(:invalid)

# Raises RuntimeError: "Request failed with status 404: ..."
Client.read!("cus_nonexistent")
ID Prefix Recognition
The client automatically recognizes Stripe ID prefixes:
	cus_* → /customers/{id}
	product_* → /products/{id}
	price_* → /prices/{id}
	sub_* → /subscriptions/{id}
	inv_* → /invoices/{id}
	evt_* → /events/{id}
	cs_* → /checkout/sessions/{id}

Examples
# Fetch a customer
{:ok, response} = Client.read("cus_123")
customer = response.body

# List customers with pagination
{:ok, response} = Client.read(:customers, limit: 10, starting_after: "cus_123")
customers = response.body["data"]

# Create a customer
{:ok, response} = Client.create(:customers, %{
  email: "customer@example.com",
  name: "Jane Doe"
})

# Update a customer
{:ok, response} = Client.update("cus_123", %{
  metadata: %{user_id: "12345"}
})

# Delete a customer
{:ok, response} = Client.delete("cus_123")

    Functions
  


    
      
        create(entity, params, options \\ [])

      


        Creates a new resource.



    


    
      
        create!(entity, params, options \\ [])

      


        Creates a new resource, raising on error.



    


    
      
        delete(id, options \\ [])

      


        Deletes a resource by ID.



    


    
      
        delete!(id, options \\ [])

      


        Deletes a resource by ID, raising on error.



    


    
      
        new(options \\ [])

      


    


    
      
        read(id_or_entity, options \\ [])

      


        Reads a resource by ID or lists resources by entity type.



    


    
      
        read!(id_or_entity, options \\ [])

      


        Reads a resource by ID or lists resources by entity type, raising on error.



    


    
      
        request(url, options \\ [])

      


    


    
      
        request!(url, options \\ [])

      


    


    
      
        update(id, params, options \\ [])

      


        Updates an existing resource by ID.



    


    
      
        update!(id, params, options \\ [])

      


        Updates an existing resource by ID, raising on error.

Creates a new resource.
Parameters
	entity - Atom representing the entity type (e.g., :customers, :products)
	params - Map of parameters for the resource
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(TinyElixirStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_new", email: "new@example.com"})
...> end)
iex> {:ok, response} = TinyElixirStripe.Client.create(:customers, %{email: "new@example.com"})
iex> response.body["id"]
"cus_new"

# Unrecognized entity type
iex> TinyElixirStripe.Client.create(:invalid, %{})
{:error, :unrecognized_entity_type}

  



    

  
    
      
    
    
      create!(entity, params, options \\ [])



        
          
        

    

  


  

Creates a new resource, raising on error.
Similar to create/3 but raises a RuntimeError on error instead of returning an error tuple.
Parameters
	entity - Atom representing the entity type
	params - Map of parameters for the resource
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(TinyElixirStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_new"})
...> end)
iex> response = TinyElixirStripe.Client.create!(:customers, %{email: "test@example.com"})
iex> response.body["id"]
"cus_new"

  



    

  
    
      
    
    
      delete(id, options \\ [])



        
          
        

    

  


  

Deletes a resource by ID.
Parameters
	id - String ID of the resource (e.g., "cus_123")
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(TinyElixirStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", deleted: true})
...> end)
iex> {:ok, response} = TinyElixirStripe.Client.delete("cus_123")
iex> response.body["deleted"]
true

Deletes a resource by ID, raising on error.
Similar to delete/2 but raises a RuntimeError on error instead of returning an error tuple.
Parameters
	id - String ID of the resource
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(TinyElixirStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", deleted: true})
...> end)
iex> response = TinyElixirStripe.Client.delete!("cus_123")
iex> response.body["deleted"]
true

Reads a resource by ID or lists resources by entity type.
Parameters
	id_or_entity - Either a string ID (e.g., "cus_123") or an atom entity type (e.g., :customers)
	options - Keyword list of options (e.g., query parameters for list operations)

Examples
# Fetch by ID
iex> Req.Test.stub(TinyElixirStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", email: "test@example.com"})
...> end)
iex> {:ok, response} = TinyElixirStripe.Client.read("cus_123")
iex> response.body["id"]
"cus_123"

# List resources
iex> Req.Test.stub(TinyElixirStripe, fn conn ->
...>   Req.Test.json(conn, %{object: "list", data: []})
...> end)
iex> {:ok, response} = TinyElixirStripe.Client.read(:customers)
iex> response.body["object"]
"list"

# Unrecognized entity type
iex> TinyElixirStripe.Client.read(:invalid)
{:error, :unrecognized_entity_type}

Reads a resource by ID or lists resources by entity type, raising on error.
Similar to read/2 but raises a RuntimeError on error instead of returning an error tuple.
Parameters
	id_or_entity - Either a string ID or an atom entity type
	options - Keyword list of options

Examples
iex> Req.Test.stub(TinyElixirStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123"})
...> end)
iex> response = TinyElixirStripe.Client.read!("cus_123")
iex> response.body["id"]
"cus_123"

Updates an existing resource by ID.
Parameters
	id - String ID of the resource (e.g., "cus_123")
	params - Map of parameters to update
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(TinyElixirStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", name: "Updated"})
...> end)
iex> {:ok, response} = TinyElixirStripe.Client.update("cus_123", %{name: "Updated"})
iex> response.body["name"]
"Updated"

  



    

  
    
      
    
    
      update!(id, params, options \\ [])



        
          
        

    

  


  

Updates an existing resource by ID, raising on error.
Similar to update/3 but raises a RuntimeError on error instead of returning an error tuple.
Parameters
	id - String ID of the resource
	params - Map of parameters to update
	options - Keyword list of additional Req options

Examples
iex> Req.Test.stub(TinyElixirStripe, fn conn ->
...>   Req.Test.json(conn, %{id: "cus_123", name: "Updated"})
...> end)
iex> response = TinyElixirStripe.Client.update!("cus_123", %{name: "Updated"})
iex> response.body["name"]
"Updated"

  


        

      


  

    
TinyElixirStripe.WebhookController 
    



      
Base controller for handling Stripe webhook events.
This module provides a use macro that injects webhook handling functionality
into your Phoenix controller. It automatically verifies webhook signatures and
forwards events to your handler module.
Usage
Create a controller in your Phoenix app:
defmodule MyAppWeb.StripeWebhookController do
  use TinyElixirStripe.WebhookController, 
    handler: MyApp.StripeWebhookHandlers
end
Then add it to your router:
scope "/webhooks" do
  pipe_through [:api]

  post "/stripe", StripeWebhookController, :create
end
Configuration
Configure your webhook secret:
config :tiny_elixir_stripe,
  stripe_webhook_secret: "whsec_..."
Security
This controller automatically verifies webhook signatures using the
stripe-signature header. Invalid signatures are rejected with a 400 response.
The raw request body must be available in conn.assigns.raw_body for signature
verification to work. Use TinyElixirStripe.ParsersWithRawBody in your endpoint.
Handler Module
The handler module should use TinyElixirStripe.WebhookHandler and define
event handlers:
defmodule MyApp.StripeWebhookHandlers do
  use TinyElixirStripe.WebhookHandler

  handle "customer.created", fn event ->
    # Process customer.created event
    :ok
  end

  handle "invoice.paid", fn event ->
    # Process invoice.paid event
    :ok
  end
end

      




  

    
TinyElixirStripe.WebhookHandler 
    



      
Spark DSL for defining webhook handlers.
This module provides a DSL for defining handlers for Stripe webhook events.
Example
defmodule MyApp.StripeWebhookHandlers do
  use TinyElixirStripe.WebhookHandler

  handle "customer.created", fn event ->
    # Process customer.created event
    :ok
  end

  handle "charge.succeeded", MyApp.ChargeHandler
end
Handlers can be either:
	Anonymous functions that take the event as an argument
	Module names that implement a handle_event/1 function

Options
	:extensions (list of module that adopts Spark.Dsl.Extension) - A list of DSL extensions to add to the Spark.Dsl

	:otp_app (atom/0) - The otp_app to use for any application configurable options

	:fragments (list of module/0) - Fragments to include in the Spark.Dsl. See the fragments guide for more.



      




  

    
TinyElixirStripe.WebhookSignature 
    



      
Verifies Stripe webhook signatures.
Stripe signs webhook events and includes the signature in the stripe-signature
header. This module verifies that signature to ensure the webhook came from Stripe.
See: https://docs.stripe.com/webhooks#verify-official-libraries
Configuration
Configure your webhook secret in your application config:
config :tiny_elixir_stripe,
  stripe_webhook_secret: "whsec_..."
Usage
payload = ~s({"id": "evt_test", "type": "customer.created"})
signature = get_req_header(conn, "stripe-signature")
secret = Application.fetch_env!(:tiny_elixir_stripe, :stripe_webhook_secret)

case WebhookSignature.verify(payload, signature, secret) do
  :ok ->
    # Process webhook
  {:error, reason} ->
    # Reject webhook
end

        Signs payload with timestamp and secret.



    


    
      
        verify(payload, signature, secret)

      


        Verifies payload against signature and secret.

          @spec sign(binary(), integer(), binary()) :: binary()


      


Signs payload with timestamp and secret.
Useful for testing webhook handling.
Examples
iex> payload = ~s({"type": "test"})
iex> timestamp = System.system_time(:second)
iex> secret = "whsec_test"
iex> signature = TinyElixirStripe.WebhookSignature.sign(payload, timestamp, secret)
iex> String.starts_with?(signature, "t=")
true

  



  
    
      
    
    
      verify(payload, signature, secret)



        
          
        

    

  


  

      

          @spec verify(binary(), binary(), binary()) :: :ok | {:error, binary()}


      


Verifies payload against signature and secret.
Returns :ok if the signature is valid and not expired.
Returns {:error, reason} otherwise.
The signature must be less than 300 seconds (5 minutes) old to prevent
replay attacks.
Examples
iex> payload = ~s({"type": "test"})
iex> timestamp = System.system_time(:second)
iex> secret = "whsec_test"
iex> signature = TinyElixirStripe.WebhookSignature.sign(payload, timestamp, secret)
iex> TinyElixirStripe.WebhookSignature.verify(payload, signature, secret)
:ok

  


        

      


  

    
TinyElixirStripe.ParsersWithRawBody 
    



      
A custom Plug.Parsers that caches the raw request body for webhook routes.
This plug conditionally caches the raw body only for webhook endpoints
(default: /webhooks/stripe). For all other routes, it behaves like standard
Plug.Parsers.
The raw body is needed for webhook signature verification, as the signature
is computed over the exact bytes received from Stripe.
Usage
In your Phoenix endpoint, replace Plug.Parsers with this module:
# Before:
plug Plug.Parsers,
  parsers: [:urlencoded, :multipart, :json],
  pass: ["*/*"],
  json_decoder: Phoenix.json_library()

# After:
plug TinyElixirStripe.ParsersWithRawBody,
  parsers: [:urlencoded, :multipart, :json],
  pass: ["*/*"],
  json_decoder: Phoenix.json_library()
The raw body will be available in conn.assigns.raw_body as a list of
binary chunks for webhook routes.
Configuration
The default webhook path is ["webhooks", "stripe"] which matches the URL
path /webhooks/stripe. This will be configurable via Igniter in future versions.

        cache_raw_body(conn, opts)

      


        Custom body reader that caches the raw request body.

Custom body reader that caches the raw request body.
This function is passed to Plug.Parsers via the :body_reader option.
It reads the body and stores it in conn.assigns.raw_body as a list
of chunks (prepended for efficiency).

  


        

      


  

    
TinyElixirStripe.WebhookHandler.Dsl 
    



      
DSL extension for defining webhook handlers.

      




  

    
TinyElixirStripe.WebhookHandler.Info 
    



      
Introspection functions for webhook handler DSL.

          @spec handlers(dsl_or_extended :: module() | map()) :: [struct()]


      


handlers DSL entities

  


        

      


  

    
mix tiny_elixir_stripe.gen.handler 
    



      
Generate a Stripe webhook event handler
Generates a handler for a specific Stripe webhook event.
Example
mix tiny_elixir_stripe.gen.handler customer.created

Arguments
	event - The Stripe event name (e.g., "customer.created", "charge.succeeded")

Options
	--handler-type - Type of handler to generate: "function" or "module" (default: prompts user)
	--module - Custom module name for module handlers (default: auto-generated from event name)
	--create-handler-module - Module name to create if no WebhookHandler module exists

Handler Types
Function Handler
Generates an inline function handler in your WebhookHandler module:
handle "customer.created", fn event ->
  # Handle customer.created event
  :ok
end
Module Handler
Generates a separate module with a handle_event/1 function:
# In your WebhookHandler module
handle "customer.created", MyApp.StripeWebhookHandlers.CustomerCreated

# Generated module at lib/my_app/stripe_webhook_handlers/customer_created.ex
defmodule MyApp.StripeWebhookHandlers.CustomerCreated do
  def handle_event(event) do
    # Handle customer.created event
    :ok
  end
end

      




  

    
mix tiny_elixir_stripe.install 
    



      
Install TinyElixirStripe with Stripe webhook support
This installer will:
	Replace Plug.Parsers with TinyElixirStripe.ParsersWithRawBody in your Phoenix endpoint
	Create a stub WebhookHandler module in lib/{app} for defining event handlers
	Generate a StripeWebhookController in lib/{app}_web that uses TinyElixirStripe.WebhookController
	Add a webhook route to your router that points to the generated controller
	Add :tiny_elixir_stripe to import_deps in .formatter.exs for DSL formatting support

The ParsersWithRawBody plug caches the raw request body for webhook signature verification,
as required by Stripe's webhook security.
The generated controller automatically handles signature verification and forwards events
to your WebhookHandler module.
Example
mix tiny_elixir_stripe.install --path /webhooks/stripe

Options
	--path or -p - The webhook endpoint path (default: "/webhooks/stripe")


      




  

    
mix tiny_elixir_stripe.set_webhook_path 
    



      
Update the webhook endpoint path for TinyElixirStripe
This task updates the webhook route in your Phoenix router to use a new path.
It will find the existing StripeWebhookController route and update
it to the new path you specify.
Example
mix tiny_elixir_stripe.set_webhook_path /new/webhook/path

Arguments
	path - The new webhook path (e.g., "/webhooks/stripe" or "/api/stripe-webhook")


      




  

    
mix tiny_elixir_stripe.sync_webhook_handlers 
    



      
Sync webhook handlers with Stripe's configured webhook endpoints
This task will:
	Fetch all webhook endpoints from your Stripe account using the Stripe CLI
	Extract all enabled events from those endpoints
	Compare them with handlers in your WebhookHandler module
	Generate handlers for any missing events

This is useful for keeping your local webhook handlers in sync with your
Stripe webhook configuration, especially after adding new events in the
Stripe Dashboard.
Example
mix tiny_elixir_stripe.sync_webhook_handlers

Options
	--api-key or -k - The Stripe API key to use (prompts to use config key if not provided)
	--handler-type or -t - Type of handler to generate: "function", "module", or "ask" (default: prompts user)
	--skip-confirmation or -y - Skip confirmation prompts and generate all missing handlers
	--create-handler-module - Module name to create if no WebhookHandler module exists

Prerequisites
This task requires the Stripe CLI to be installed and authenticated:
# Install Stripe CLI (macOS)
brew install stripe/stripe-cli/stripe

# Login to Stripe
stripe login

Handler Types
Function Handler (inline)
Generates inline function handlers in your WebhookHandler module:
handle "customer.created", fn event ->
  # Handle customer.created event
  :ok
end
Module Handler
Generates separate modules with a handle_event/1 function:
# In your WebhookHandler module
handle "customer.created", MyApp.StripeWebhookHandlers.CustomerCreated

# Generated module
defmodule MyApp.StripeWebhookHandlers.CustomerCreated do
  def handle_event(event) do
    # Handle customer.created event
    :ok
  end
end
Ask (interactive)
Prompts you for each event individually, allowing you to choose
function or module handlers on a per-event basis.

      




  

    
mix tiny_elixir_stripe.update_supported_events 
    



      
Updates the list of supported Stripe webhook events.
This task is for library contributors to update the list of valid Stripe events
that can be used with the webhook handler generators.
Usage
mix tiny_elixir_stripe.update_supported_events
This task will:
	Run stripe trigger --help to get the list of supported events
	Parse the output to extract event names
	Write the events to priv/supported_stripe_events.txt

Requirements
You must have the Stripe CLI installed and available in your PATH.
Install it from: https://stripe.com/docs/stripe-cli

